
0930        .LIST OFF 

0940 ;-------------------------------- 

0950 ;  DOS 3.3 BOOT ROM 

0960 ; 

0970 ;    APPLE ASSEMBLY LINE 

0980 ;      V1N10P17, AUG 1981 

0990 ; 

1000 ;-------------------------------- 

1010 ;  DOS 3.3 BOOT ROM $C600-$C6FF 

1020 ;  BOB SANDER-CEDERLOF   4/7-81 

1030 ;-------------------------------- 

1040 ;       DISK CONTROLLER ADDRESSES 

1050 ;-------------------------------- 

1060 PHASE.OFF           .EQ $C080 

1070 PHASE.ON            .EQ $C081 

1080 MOTOR.OFF           .EQ $C088 

1090 MOTOR.ON            .EQ $C089 

1100 ENABLE.DRIVE.1      .EQ $C08A 

1110 ENABLE.DRIVE.2      .EQ $C08B 

1120 Q6L                 .EQ $C08C 

1130 Q6H                 .EQ $C08D 

1140 Q7L                 .EQ $C08E 

1150 Q7H                 .EQ $C08F 

1160 ; 

1170 ;      Q6   Q7    USE OF Q6 AND Q7 LINES 

1180 ;     ---- ----   ---------------------- 

1190 ;     LOW  LOW    READ (DISK TO SHIFT REGISTERS) 

1200 ;     LOW  HIGH   WRITE (SHIFT REGISTER TO DISK) 

1210 ;     HIGH LOW    SNESE WRITE PROTECT 

1220 ;     HIGH HIGH   LOAD SHIFT REGISTER FROM DATA BUS 

1230 ;-------------------------------- 

1240 BUF.PNTR    .EQ $26,27 

1250 SLOT16      .EQ $2B    SLOT*16 

1260 SECTOR      .EQ $3D 

1270 TRACK       .EQ $41 

1280 STACK       .EQ $100 

1290 POST.NYBBLE.CODES .EQ $2D6 

1300 LITTLE.BUFFER     .EQ $300 

1310 MON.RTS     .EQ $FF58 

1320 MON.WAIT    .EQ $FCA8 

1330 ;-------------------------------- 

1340         .OR $C600 

1350         .T0 

1360 ;-------------------------------- 

1370 BOOT.3.3 

1380         LDX #$20    REDUNDANT INSTRUCTION, USED 

1390 ;                   TO IDENTIFY CONTROLLER CARD 

1400 ;-------------------------------- 

1410 ;       GENERATE POST-NYBBLE CONVERSION TABLE 

1420 ;           FILLS IN THOSE SLOTS WHOSE INDEX 

1430 ;           RELATIVE TO POST.NYBBLE.CODES IS 

1440 ;           A VALID NYBBLE CODE. (VALID CODES 

1450 ;           HAVE AT MOST ONE PAIR OF ADJACENT 

1460 ;           0-BITS, AND AT LEAST ONE PAIR OF 

1470 ;           ADJACENT 1-BITS IN BITS 0-6.) 

1480 ;-------------------------------- 

1490         LDY #0 



1500         LDX #3      COULD BE ANY VALUE FROM 0 TO $16 

1510 ;                   3 USED FOR CONTROLLER ID 

1520 .1      STX $3C     CHECK CODE FOR VALID NYBBLE 

1530         TXA  

1540         ASL  

1550         BIT $3C     TEST (X .AND. 2*X) 

1560         BEQ .3      NO ADJACENT 1-BITS, NO GOOD 

1570         ORA $3C     TEST ADJACENT 0-BITS 

1580         EOR #$FF    CHANGE 1'S FOR TEST 

1590         AND #$7E    DON'T CARE ABOUT BIT 7 

1600 .2      BCS .3      NOT VALID NYBBLE CODE 

1610         LSR  

1620         BNE .2 

1630         TYA  

1640         STA POST.NYBBLE.CODES+$80,X 

1650         INY  

1660 .3      INX  

1670         BPL .1 

1680 ;-------------------------------- 

1690         JSR MON.RTS GET THIS LOCATION ON STACK 

1700         TSX         FIND PAGE BYTE ON STACK 

1710         LDA STACK,X 

1720         ASL         ISOLATE SLOT # 

1730         ASL         AND MULTIPLY BY 16 

1740         ASL  

1750         ASL  

1760         STA SLOT16  SLOT # TIMES 16 

1770         TAX  

1780         LDA Q7L,X   SET UP TO READ DRIVE 

1790         LDA Q6L,X 

1800         LDA ENABLE.DRIVE.1,X 

1810         LDA MOTOR.ON,X 

1820 ;-------------------------------- 

1830 ;       MOVE TO TRACK 0 (ASSUME WORST CASE 

1840 ;       INITIAL POSITION OF TRACK 40) 

1850 ;-------------------------------- 

1860         LDY #80     80 HALF-TRACKS 

1870 .4      LDA PHASE.OFF,X  STEPPER MOTOR PHASE OFF 

1880         TYA         COMPUTE NEXT PHASE 

1890         AND #3      YIELDS 3,2,1,0 

1900         ASL         YIELDS 6,4,2,0 

1910         ORA SLOT16  MERGE WITH SLOT*16 

1920         TAX  

1930         LDA PHASE.ON,X  STEPPER MOTOR PHASE ON 

1940         LDA #86     WAIT 19.2 MILLISECONDS 

1950         JSR MON.WAIT NO CHANGE TO X OR Y, A=0 

1960         DEY  

1970         BPL .4 

1980 ;-------------------------------- 

1990 ;       A=0, X=SLOT*16 

2000 ;-------------------------------- 

2010         STA BUF.PNTR  ($00 -> LOW BYTE OF PNTR) 

2020         STA SECTOR 

2030         STA TRACK 

2040         LDA #8      BUFFER AT $800 

2050         STA BUF.PNTR+1  ($08 -> HI-BYTE OF PNTR) 

2060 ;-------------------------------- 



2070 READ.SECTOR 

2080 .1      CLC         FLAG CLEAR, LOOK FOR $D5 AA 96 

2090 .2      PHP         SAVE FLAG ON STACK 

2100 .3      LDA Q6L,X   READ DISK 

2110         BPL .3 

2120 .4      EOR #$D5 

2130         BNE .3      NO 

2140 .5      LDA Q6L,X   READ DISK 

2150         BPL .5 

2160         CMP #$AA 

2170         BNE .4 

2180         NOP  

2190 .6      LDA Q6L,X   READ DISK 

2200         BPL .6 

2210         CMP #$96 

2220         BEQ .7      FOUND ADDRESS MARK: $D5 AA 96 

2230         PLP         RETRIEVE FLAG 

2240         BCC .1      LOOKING FOR ADDRESS HEADER 

2250         EOR #$AD    LOOKING FOR DATA HEADER 

2260         BEQ FILL.BUFFER 

2270         BNE .1      START ALL OVER 

2280 ;-------------------------------- 

2290 .7      LDY #3      READ VOLUME, TRACK, SECTOR 

2300 .8      STA $40 

2310 .9      LDA Q6L,X   READ DISK 

2320         BPL .9 

2330         ROL         SAVE UPPER SLICE 

2340         STA $3C 

2350 .10     LDA Q6L,X   READ DISK 

2360         BPL .10 

2370         AND $3C     MERGE SLICES 

2380         DEY         3RD BYTE YET? 

2390         BNE .8      NO, GET ANOTHER 

2400         PLP         THROW AWAY FLAG 

2410         CMP SECTOR  CORRECT SECTOR? 

2420         BNE .1      NO 

2430         LDA $40     CORRECT TRACK? 

2440         CMP TRACK 

2450         BNE .1      NO 

2460         BCS .2      YES, SET FLAG FOR DATA HEADER 

2470 ;                   AND BRANCH BACK ALWAYS 

2480 ;-------------------------------- 

2490 ;       A=0 ON ENTRY 

2500 ;-------------------------------- 

2510 FILL.BUFFER 

2520         LDY #86     READ 86 BYTES 

2530 .1      STY $3C 

2540 .2      LDY Q6L,X   READ BYTE 

2550         BPL .2 

2560         EOR POST.NYBBLE.CODES,Y  DECODE BYTE 

2570         LDY $3C 

2580         DEY  

2590         STA LITTLE.BUFFER,Y 

2600         BNE .1 

2610 ;-------------------------------- 

2620 .3      STY $3C     Y=0 

2630 .4      LDY Q6L,X   READ BYTE 



2640         BPL .4 

2650         EOR POST.NYBBLE.CODES,Y  DECODE BYTE 

2660         LDY $3C 

2670         STA (BUF.PNTR),Y 

2680         INY  

2690         BNE .3 

2700 .5      LDY Q6L,X   READ CHECKSUM BYTE 

2710         BPL .5 

2720         EOR POST.NYBBLE.CODES,Y 

2730 .6      BNE READ.SECTOR  BAD CHECKSUM, START OVER 

2740 ;-------------------------------- 

2750         LDY #0 

2760 .7      LDX #86     PATCH THE 6+2 BACK TOGETHER 

2770 .8      DEX  

2780         BMI .7      FINISHED A TRIP 

2790         LDA (BUF.PNTR),Y 

2800         LSR LITTLE.BUFFER,X 

2810         ROL  

2820         LSR LITTLE.BUFFER,X 

2830         ROL  

2840         STA (BUF.PNTR),Y 

2850         INY  

2860         BNE .8 

2870 ;-------------------------------- 

2880         INC BUF.PNTR+1  POINT AT NEXT PAGE 

2890         INC SECTOR  POINT AT NEXT SECTOR 

2900         LDA SECTOR 

2910         CMP $0800   SEE IF HAVE READ ENUF SECTORS 

2920         LDX SLOT16 

2930         BCC .6      NOT ENUF SECTORS YET 

2940         JMP $0801   GOTO REST OF BOOT 

2950 ;-------------------------------- 

2960         .HS 0000000000 


